
This case study looks at the re-engineering of a program to translate a number 
expressed as an integer into a Roman numeral. The program contains a number 
of structures which should be modified or removed to make the program more 

maintainable.

SYNOPSIS
This case study looks at the re-engineering of a program to translate a number 
expressed as an integer into a Roman numeral. The program contains a number 
of structures which should be modified or removed to make the program more 

maintainable.

Type: legacy re-engineering

Language: Fortran

Compiler: gfortran

Skills: Fortran re-engineering, re-writing legacy programs, 
language skills, removing goto’s

Experience Level: intermediate

“Design and programming are human activities; forget that and all is lost.”
Bjarne Stroustrup.

PROGRAM DESCRIPTION

The program provided converts a number expressed as an integer into a Roman numeral. To 
understand how this works, let’s consider a conversion on a digit-by-digit basis. To write the 
Roman equivalent of  1896, we first convert the first digit, the 1 in the thousands position, to the 
corresponding Roman numeral M. The next digit, the 8 in the hundreds position leads to the 
Roman numeral DCCC. The 9 in the tens position gives XC, the 6 in the units position VI, 
giving the number MDCCCXCVI. Note that a zero in any position adds nothing to the Roman 
numeral, so 107 becomes CVII. 

ALGORITHM

The algorithm works as follows:
1. Start with an empty pattern.
2. Convert thousands digit and add to pattern.
3. Convert hundreds digit and add to pattern.
4. Convert tens digit and add to pattern.
5. Convert units digit and add to pattern.

The algorithm assumes that no number will have more than four digits - say the numbers 
represent years. 

Digit Conversion
When converting a digit, there are 10 possibilities, one for each digit 0 - 9. Each produces a 
pattern of  Roman symbols, and the pattern depends on the position of  the digit in the number. A 
table of  digit conversion patterns is shown below:

thousands hundreds tens units

0  -
1  M
2  MM
3  MMM
4  MW
5  W
6  WM
7  WMM
8  WMMM
9  MY

0  -
1  C
2  CC
3  CCC
4  CD
5  D
6  DC
7  DCC
8  DCCC
9  CM

0  -
1  X
2  XX
3  XXX
4  XL
5  L
6  LX
7  LXX
8  LXXX
9  XC

0  -
1  I
2  II
3  III
4  IV
5  V
6  VI
7  VII
8  VIII
9  IX

Note that W and Y in the thousands column replace versions of  V and X that have bars atop 
them. Each column uses three symbols, so if  they are replaced by the three symbols, a, b, and c, 
then each column of  the table is identical. This leads to the following condensed table:

thousands hundreds tens units

a=M
b=W
c=Y

a=C
b=D
c=M

a=X
b=L
c=C

a=I
b=V
c=X

1! a! ! 4! ab! ! 7! baa
2! aa! ! 5! b! ! 8! baaa!
3! aaa! ! 6! ba! ! 9! ac

Converting a digit is as simple as looking up the appropriate entry in the condensed table of  digit 
conversions.

! Input = 0, output = nothing!
! Input = 1, output = a
! Input = 2, output = aa!
! Input = 3, output = aaa!
! Input = 4, output = ab!
! Input = 5, output = b!
! Input = 6, output = ba!
! Input = 7, output = baa!
! Input = 8, output = baaa!
! Input = 9, output = ac!

So 1896 becomes:

! a! ! baaa! ! ac! ! ba
! M! ! DCCC! ! XC! ! VI ! ! = ! MDCCCXCVI

Finding Digit Positions
To determine which digit is in which position, the modulus can be calculated. For example, to 
calculate the hundred digit, divide by 100, and find the modulus when divided by 10. For 
example 1896/100 = 18/10 = modulus 8. 

!     units digit:! MOD(I, 10)
!      tens digit:! MOD(I/10, 10)
!  hundreds digit:! MOD(I/100, 10)
! thousands digit:! MOD(I/1000, 10)

Putting the Pattern Together
So the final part of  the algorithm involves adding the pattern to the accumulating Roman 
numeral. This can be achieved by storing the symbols of  the Roman numeral in an array, one 
symbol to each element. Tracking the number of  symbols can be achieved in a counter, updating 
the value each time symbols are added to the Roman numeral. 

Explaining the Code
The given code uses a subroutine named DIGIT to do the digit conversion. DIGIT has a 
number of  inputs: (i) the digit to be converted, the three relevant Roman symbols (Hollerith 
characters), the array in which the Roman numeral is being accumulated (an array of  Hollerith 

characters), and the current length of  that array. Note that the length of  the array used to store 
the Roman numeral is 16, meaning no Roman numeral for a four-digit number can have more 
than 16 symbols. The digit 8 has the longest Roman equivalent, 4 symbols, so 8888 is the largest  
number which could be converted. 

THE CODE

     PROGRAM roman_number 

C     Input: integer
C     Output: input integer and its Roman equivalent
C     Variables: i -- input integer
C            r -- array to store Roman numeral
C            n -- current number of symbols in r
      INTEGER i, r(16), n, j

      WRITE(*,*) 'Enter a year '
      READ(*,100) i
  100 FORMAT(I4)

      IF (i .LE. 0) GO TO 200
C     Start with an empty Roman numeral
      n=0
C     Convert 1000's digit, 100's digit, tens, units
      CALL digit(mod(i/1000,10), 1HM, 1HW, 1HY, R, N);
      CALL digit(mod(i/100 ,10), 1HC, 1HD, 1HM, R, N);
      CALL digit(mod(i/10  ,10), 1HX, 1HL, 1HC, R, N);
      CALL digit(mod(i     ,10), 1HI, 1HV, 1HX, R, N);
      WRITE(*,101) i, (r(j),j=1,n)
  101 FORMAT('The Roman numeral for', I5,' is '31A1)
      STOP 

  200 WRITE(*,201) i
  201 FORMAT('No Roman numeral for',I5)
  
      STOP 
      END 

      SUBROUTINE digit(d, a, b, c, r, n)
      INTEGER d, a, b, c, r(31), n
C     Input: 
C            d -- number between 0 and 9
C            a,b,c -- Roman symbols
C            r -- array to store Roman numeral
C            n -- current number of symbols in r 
C     Output: 
C            r,n -- updated by converting d to Roman symbols 
      INTEGER case
      case = d + 1
      GO TO (300,301,302,303,304,305,306,307,308,309), case
C     input d=0, output pattern=nothing
  300   RETURN 
C     input d=1, output pattern=A
  301   r(n+1) = A
        n = n + 1
        RETURN 
C     input d=2, output pattern=AA

  302   r(n+1) = A
        r(n+2) = A
        n = n + 2
        RETURN 
C     input d=3, output pattern=AAA
  303   r(n+1) = A
        r(n+2) = A
        r(n+3) = A
        n = n + 3
        RETURN 
C     input d=4, output patter=AB
  304   r(n+1) = A
        r(n+2) = B
        n = n + 2
        RETURN 
C     input d=5, output patter=B
  305   r(n+1) = B
        n = n + 1
        RETURN 
C     input d=6, output patter=BA
  306   r(n+1) = B
        r(n+2) = A
        n = n + 2
        RETURN 
C     input d=6, output patter=BAA
  307   r(n+1) = B
        r(n+2) = A
        r(n+3) = A
        n = n + 3
        RETURN 
C     input d=6, output patter=BAAA
  308   r(n+1) = B
        r(n+2) = A
        r(n+3) = A
        r(n+4) = A
        n = n + 4
        RETURN 
C     input d=6, output patter=AC
  309   r(n+1) = A
        r(n+2) = C
        n = n + 2
         RETURN
      END

DISASSEMBLE AND ANALYZE THE PROGRAM

The first task is to assess the program, by disassembling it. Assume no prior knowledge, the task 
here is to re-engineer the program, not create it from scratch. Don’t be fooled by the programs 
mere 94 lines of  code. Re-engineering a program can be a challenging experience, no matter the 
size of  the program.

Some basic observations can be made. Column 1 contains a C to denote comments, and 
statement numbers in columns 1-5 to be used in jump instructions for transferring control. 

Try to compile this and it will compile... although it does spit out nine warnings, all to do with 
the call to the subroutine digit. Over half  the warnings have to do with the use of  Hollerith 
constants.

! Warning: Legacy Extension: Hollerith constant at (1)

This is not a challenging program to re-engineer, and the process can be broken down into a 
number of  steps.

1. Re-engineer the simple things - convert the comment delimiters, convert the program to 
lowercase and rationalize the indenting.

2. Convert the type declarations, convert the subroutine parameters, and remove the Hollerith 
constants.

3. Remove the unconditional go to statement in the main program.
4. Remove computed go to statement in the subroutine digit.

For original code see listing romanN.for

STEP 1: THE SIMPLE FIXES

Firstly, the program as it stands has a weird mix of  lowercase and uppercase. Convert it all to 
lowercase. Also remove tabs and rationalize the indentation. Next deal with the comment 
delimiters, which are C in older renditions of  Fortran, and ! in modern versions. This basically 
means replacing this:

C     Input: integer
C     Output: input integer and its Roman equivalent
C     Variables: i -- input integer
C            r -- array to store Roman numeral
C            n -- current number of symbols in r

with this:

! Input: integer
! Output: input integer and its Roman equivalent
! Variables: i -- input integer
!            r -- array to store Roman numeral
!            n -- current number of symbols in r

See listing romanN_1.f95

STEP 2: CONVERSIONS

The next part of  the transformation deals with converting the type declarations to F95 
specifications. This basically means converting this:

    integer i, r(16), n, j

to this:

    integer :: i, n, j
    character, dimension(31) :: r

Notice how the variable r turned from an integer into a character? Early versions of  Fortran had 
no character data types, just numeric ones. Hollerith constants (named after Herman Hollerith) 
were used to store character data. So the conversion moves r from being a Hollerith constant to 
a character array. Once the Hollerith constants have been modified, the method of  specifying 
characters has to be modified as well - in the calls to subroutine digit. The original call has the 
form:

    call digit(mod(i/1000,10), 1HM, 1HW, 1HY, r, n)

where 1HM specifies the character “M”. This is a simple modification of  the form:

    call digit(mod(i/1000,10), 'M', 'W', 'Y', r, n)

Finally the parameters within the subroutine digit are modified. This means converting this:

subroutine digit(d, a, b, c, r, n)
    integer d, a, b, c, r(31), n

to this:

subroutine digit(d, a, b, c, r, n)
implicit none
    ! Declare calling parameter types and definitions
    integer, intent(in) :: d
    character, intent(in) :: a, b, c
    character, intent(inout), dimension(31) :: r
    integer, intent(inout) :: n

Part of  the conversion requires specifying whether the parameter intent is in, out or inout. 
The other part involves converting the variables a, b, c, and r from integers to characters. The 
use of  intent allows the compiler to provide additional checking.

See listing romanN_2.f95

STEP 3: REMOVE UNCONDITIONAL GOTO

The third step involves removing the unconditional goto statement in the main program. Here is  
the block of  code containing the goto statement:

    if (i .le. 0) go to 200
    n=0

    call digit(mod(i/1000,10), 'M', 'W', 'Y', r, n)
    call digit(mod(i/100 ,10), 'C', 'D', 'M', r, n)
    call digit(mod(i/10  ,10), 'X', 'L', 'C', r, n)
    call digit(mod(i     ,10), 'I', 'V', 'X', r, n)

    write(*,101) i, (r(j),j=1,n)
    
101 format('The Roman numeral for', I5,' is '31A1)
    stop

200 write(*,201) i
201 format('No Roman numeral for',I5)

This code basically jumps to code associated with label 200 if  i (the year entered) is less than or 
equal to zero. Otherwise the program continues running, stopping when the Roman numeral has  
been calculated and output. Fixing this simply involves swapping the logic around and adding an 
else statement. Here is the re-engineered code:

    if (i .le. 0) then
        write(*,201) i
        201 format('No Roman numeral for',I5)
    else
        ! Start with an empty Roman numeral
        n=0

        ! Convert 1000's digit, 100's digit, tens, units
        ! Hollerith constants are removed by replacing with character
        ! constants e.g. 1HM becomes 'M'
        call digit(mod(i/1000,10), 'M', 'W', 'Y', r, n)
        call digit(mod(i/100 ,10), 'C', 'D', 'M', r, n)
        call digit(mod(i/10  ,10), 'X', 'L', 'C', r, n)
        call digit(mod(i     ,10), 'I', 'V', 'X', r, n)

        write(*,101) i, (r(j),j=1,n)
    101 format('The Roman numeral for', I5,' is '31A1)

    end if

The goto is gone!

See listing romanN_3.f95

STEP 4: REMOVE COMPUTED GOTO

The final re-engineering step involves removing the computed goto statement in the subroutine 
digit. Here is part of  the code for the computed goto statement:

    go to (300,301,302,303,304,305,306,307,308,309), case

    ! input d=0, output pattern=nothing
300 return

    ! input d=1, output pattern=A
301 r(n+1) = a
    n = n + 1
    return
  
    ...

309 r(n+1) = a
    r(n+2) = c
    n = n + 2
    return

This code basically jumps to code associated with label X, where X is value of  the variable case.  
So if  case has the value 8, then the computed goto jumps to label 307 (because it is the 8th label 
in the statement). There are two ways of  fixing this: use a  (i) case structure, or (ii) if-else 
sequence. Here is the re-engineered code using a case structure:

    select case(cR)
    case(1)
    ! input d=0, output pattern=nothing
    case(2)
    ! input d=1, output pattern=A
        r(n+1) = a
        n = n + 1
        return

    ...

    case(10)
    ! input d=9, output patter=AC
        r(n+1) = a
        r(n+2) = c
        n = n + 2
        return
    end select

This removes the final legacy issue from the program.

See listing romanN_4.f95

REENGINEERING - FINAL

What is left to do? Only one this really stands out, and that is changing the variable names to 
something more meaningful. There are two variables in the main program which should be 
modified: i and r. The variable i represents the year input by the user and should be changed to 
year. The variable r represents the character array which stored the derived Roman numeral - 
this can be changed to romanN.

See the final listing romanN_5.f95

TEST THE CODE

Does it still work?

! Enter a year
! 1957
! The Roman numeral for 1957 is MCMLVII

Note that testing the code is imperative all the way through the process of  re-engineering. You can’t 
just re-engineer everything and expect it to work. Try and create a new file each time a major re-
engineer has finished (and tested) - this way if  something goes wrong, there is a backup of  the re-
engineered components. 

M.Wirth © 2014


